module mod_two_dim_merge_sort
  implicit none
  integer, private, parameter :: intkind = 4
contains
  logical function less(a, b) result(res)
    integer(intkind), intent(in) :: a(:), b(:)
    res = a(1) > b(1) .or. (a(1) == b(1) .and. a(2) >= b(2))
  end
  subroutine merge_sort(a)
    integer(intkind), intent(inout) :: a(:, :)
    integer :: n, m, l, i, u
    n = size(a, 1)
    m = n
    l = 1
    do while (m > 1)
      do i = 1, m / 2
        u = min(2 * i * l, n)
        call merger(a(2 * (i - 1) * l + 1:(2 * i - 1) * l, :), &
        a((2 * i - 1) * l + 1:u, :))
      end do
      l = 2 * l
      m = (m + 1) / 2
    end do
  end
  subroutine merger(a1, a2)
    integer(intkind), intent(inout) :: a1(:, :), a2(:, :)
    integer(intkind) :: a(size(a1, 1) + size(a2, 1), size(a1, 2))
    integer :: i1, i2, n1, n2
    i1 = 1
    i2 = 1
    n1 = size(a1,1)
    n2 = size(a2,1)
    do while (i1 <= n1 .and. i2 <= n2)
      if (less(a1(i1, :), a2(i2, :))) then
        a(i1 + i2 - 1, :) = a1(i1, :)
        i1 = i1 + 1
      else
        a(i1 + i2 - 1, :) = a2(i2, :)
        i2 = i2 + 1
       end if
    end do
    if (i1 <= n1) then
      a(i1 + i2 - 1:n1 + n2, :) = a1(i1:n1, :)
    else
      a(i1 + i2 - 1:n1 + n2, :) = a2(i2:n2, :)
    end if
    a1 = a(1:n1, :)
    a2 = a(n1 + 1:n1 + n2, :)
  end
end module mod_two_dim_merge_sort
program sum_of_large_numbers
  use mod_two_dim_merge_sort
  implicit none
  integer :: n, a(2000, 2) = 0, i, j
  integer(8) :: dp(0:2000, 0:2000) = 0, x = 0
  read(*,*) n
  read(*,*) a(1:n, 1)
  a(1:n, 2) = [(i, i = 1, n)]
  call merge_sort(a(1:n, :))
  do i = 0, n
    do j = 0, n - i
      if (i > 0) then
        dp(i, j) = max(dp(i, j), &
        dp(i - 1, j) + a(i + j, 1) * abs(a(i + j, 2) - i))
      end if
      if (j > 0) then
        dp(i, j) = max(dp(i, j), &
        dp(i, j - 1) + a(i + j, 1) * abs(a(i + j, 2) - (n - j + 1)))
      end if
    end do
  end do
  do i = 0, n
    x = max(x, dp(i, n - i))
  end do
  write(*,'(i0)') x
end program sum_of_large_numbers